Nehomogeninės lygties sprendimas

Svarbiausi uždavinio sprendimo žingsniai:

a) diferencialinės lygties užrašymas,
b) homogeninės lygties sprendinio radimas,
c) atskirojo sprendinio pavidalo nustatymas,
d) neapibrėžtinių konstantų nustatymas,
e) bendrojo sprendinio sudarymas,
f) atskirojo sprendinio užrašymas,
g) atskirojo sprendinio grafinis pavaizdavimas.

Nagrinėsime taikymuose dažniau sutinkamas antrosios eilės tiesines nehomogenines diferencialines lygtis su pastoviais koeficientais.

Spręskime diferencialinę lygtį:

$$y'' + 4y = -8\sin 2x + 32\cos 2x + 4e^{2x}$$

```
> restart;
```

Užrašome duotą diferencialinę lygtį:

```
> d_lygtis:=diff(y(x),x,x)+4*y(x)=-8*sin(2*x)+32*
cos(2*x)+4*exp(2*x);
```

$$d\_lygtis := \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) + 4\,\mathrm{y}(x) = -8\sin(2\,x) + 32\cos(2\,x) + 4\,\mathbf{e}^{(2\,x)}$$

Sudarome ją atitinkančią homogeninę diferencialinę lygtį ir dešinės pusės funkciją:

```
> d_homogenine:=diff(y(x),x,x)+4*y(x)=0;
```

$$d\_homogenine := \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) + 4\,\mathrm{y}(x) = 0$$

```
> d_puses_f:=-8*sin(2*x)+32* cos(2*x)+4*exp(2*x);
```

$$d\_puses\_f := -8\sin(2\,x) + 32\cos(2\,x) + 4\,\mathbf{e}^{(2\,x)}$$

Užrašome charakteringąją lygtį ir randame jos šaknis:

```
> char_lygtis:= lambda^2+4=0;solve(char_lygtis,lambda);
```

$$char\_lygtis := \lambda^2 + 4 = 0$$

$$2\,I,\ -2\,I$$

Randame homogeninės diferencialinės lygties bendrąjį sprendinį:

```
> y_homogen:=dsolve(d_homogenine, y(x));
```

$$y\_homogen := \mathrm{y}(x) = \_C1\sin(2\,x) + \_C2\cos(2\,x)$$

Sudarome atskirąjį nehomogeninės diferencialinės lygties sprendinį su neapibrėžtiniais koeficientais:

kadangi nehomogeninės diferencialinės lygties dešinės pusės funkcija sudaryta iš dviejų funkcijų sumos, t.y. viena funkcija yra

```
> f_1:=-8*sin(2*x)+32* cos(2*x);
```

$$f\_1 := -8\sin(2x) + 32\cos(2x)$$

o kita –

```
> f_2:=4*exp(2*x);
```

$$f\_2 := 4\,e^{(2x)}$$,

tai Y (atskirasis nehomogeninės lygties sprendinys) bus taip pat dviejų funkcijų suma, kurių pirmoji bus su trigonometrinėmis funkcijomis, o antroji – tik su eksponentine funkcija:

```
> Y_1_atskir:=(A*sin(2*x)+B* cos(2*x))*x;
```

$$Y\_1\_atskir := (A\sin(2x) + B\cos(2x))\,x$$

```
> Y_2_atskir:=C*exp(2*x);
```

$$Y\_2\_atskir := C\,e^{(2x)}$$

Y_1 sprendinį sudarome nagrinėdami pirmąją funkciją *f_1* ir charakteringosios lygties šaknis. Iš funkcijos *f_1* pavidalo matome, kad prie trigonometrinių funkcijų yra nulinio laipsnio daugianariai, t.y. konstantos, o kompleksiniai skaičiai *2i* ir *-2i* yra pirmojo kartotinumo charakteringosios lygties šaknys.

Y_2 sprendinį sudarome nagrinėdami funkciją *f_2* ir charakteringosios lygties šaknis. Iš funkcijos *f_2* pavidalo matome, kad prie eksponentinės funkcijos yra nulinio laipsnio daugianaris, t.y. konstanta, o realusis skaičius *2* nėra charakteringosios lygties šaknis.

Tuomet atskirojo nehomogeninės diferencialinės lygties sprendinio pavidalas yra:

```
> Y_atskir:=Y_1_atskir+Y_2_atskir;
```

$$Y\_atskir := (A\sin(2x) + B\cos(2x))\,x + C\,e^{(2x)}$$

Nustatome neapibrėžtinius koeficientus:
atskirąjį sprendinį įrašome į duotą diferencialinę lygtį ir suformuojame tiesinių lygčių sistemą nežinomoms konstantoms nustatyti (sistemoje turi būti tiek lygčių, kiek yra nežinomų konstantų):

```
> lygtis:=eval(subs(y(x)=Y_atskir,d_lygtis));
```

$$lygtis := (-4A\sin(2x) - 4B\cos(2x))\,x + 4A\cos(2x) - 4B\sin(2x) + 8C\,e^{(2x)} + 4\,(A\sin(2x) + B\cos(2x))\,x = -8\sin(2x) + 32\cos(2x) + 4\,e^{(2x)}$$

```
> lygtis_1:=eval(subs(x=0, lygtis));
```

$$lygtis\_1 := 4A + 8C = 36$$

```
> lygtis_2:=eval(subs(x=1, lygtis));
```

$$lygtis\_2 := 4A\cos(2) - 4B\sin(2) + 8C\,e^2 = -8\sin(2) + 32\cos(2) + 4\,e^2$$

```
> lygtis_3:=eval(subs(x=2, lygtis));
```

$$lygtis\_3 := 4A\cos(4) - 4B\sin(4) + 8C\,e^4 = -8\sin(4) + 32\cos(4) + 4\,e^4$$

```
> sistema:={lygtis_1, lygtis_2, lygtis_3};
```

$$sistema := \{4A\cos(4) - 4B\sin(4) + 8C\,e^4 = -8\sin(4) + 32\cos(4) + 4\,e^4,\ 4A + 8C = 36,\ 4A\cos(2) - 4B\sin(2) + 8C\,e^2 = -8\sin(2) + 32\cos(2) + 4\,e^2\}$$

Išsprendžiame sudarytąją tiesinių lygčių sistemą:

```
> solve(sistema);
```

$$\{C = \frac{1}{2},\ B = 2,\ A = 8\}$$

Gautas konstantų reikšmes įrašome į atskirojo sprendinio išraišką ir sudarome bendrąjį nehomogeninės diferencialinės lygties sprendinį:

```
> Y:=subs(C = 1/2, B = 2, A = 8, Y_atskir);
```

$$Y := (8\sin(2x) + 2\cos(2x))\,x + \frac{1}{2}\mathbf{e}^{(2x)}$$

```
> y_nehomogen:=rhs(y_homogen)+Y;
```

$$y\_nehomogen := \_C1\sin(2x) + \_C2\cos(2x) + (8\sin(2x) + 2\cos(2x))\,x + \frac{1}{2}\mathbf{e}^{(2x)}$$

Rasime atskirąjį nehomogeninės diferencialinės lygties sprendinį, tenkinantį pradines sąlygas:

$$y(0) = 1,\; y'(0) = 0\,.$$

Pirmąją sąlygą įrašome į surastą nehomogeninės lygties sprendinį, o antrąją – į to sprendinio išvestinę ir sudarome tiesinių lygčių sistemą nežinomoms konstantoms nustatyti:

```
> spr_isvestine:=diff(y_nehomogen,x);
```

$$spr\_isvestine := 2\,\_C1\cos(2x) - 2\,\_C2\sin(2x) + (16\cos(2x) - 4\sin(2x))\,x + 8\sin(2x) + 2\cos(2x) + \mathbf{e}^{(2x)}$$

```
> l_1:=eval(subs(x=0,y_nehomogen=1));
```

$$l\_1 := \frac{1}{2} + \_C2 = 1$$

```
> l_2:=eval(subs(x=0,spr_isvestine=0));
```

$$l\_2 := 2\,\_C1 + 3 = 0$$

```
> sist:={l_1,l_2};
```

$$sist := \{2\,\_C1 + 3 = 0,\; \frac{1}{2} + \_C2 = 1\}$$

Išsprendžiame šią sistemą

```
> solve(sist);
```

$$\{\_C2 = \frac{1}{2},\; \_C1 = \frac{-3}{2}\}$$

ir gautas konstantų reikšmes įrašome į bendrąjį nehomogeninės diferencialinės lygties sprendinį:

```
> atskirasis_sprend:=subs(_C2 = 1/2, _C1 = -3/2,y_nehomogen);
```

$$atskirasis\_sprend := -\frac{3}{2}\sin(2x) + \frac{1}{2}\cos(2x) + (8\sin(2x) + 2\cos(2x))\,x + \frac{1}{2}\mathbf{e}^{(2x)}$$

Pasitikriname su komanda *dsolve:*

```
> dsolve({d_lygtis,y(0)=1,D(y)(0)=0});
>
```

$$y(x) = -\frac{3}{2}\sin(2x) - \frac{7}{2}\cos(2x) + \frac{1}{2}(8 + 4x)\cos(2x) + 8\sin(2x)\,x + \frac{1}{2}\mathbf{e}^{(2x)}$$

Nubrėžiame atskirojo sprendinio grafiką:

```
> plot(atskirasis_sprend,x=-2..2);
```